
                            

                       IDL Tutorial –SolPhys2010 

IDL is a software for data analysis, visualization and cross-platform application 
developments. It integrates a powerful, array-oriented language with numerous 
mathematical analysis and display technique, giving high flexibility. IDL allows us to exploit 
interactively through integrated graphical facilities large data sets.  Among the main 
characteristics are: 

INTRODUCTION 

• Multi-platform 
• Array oriented 
• Graphic visualisation: plots, contours, animation 
• Mathematical and statistical tools 
• Flexible file reading/writing 
• Conexion with Fortran and C 

 

You can choose to work with IDL using the command line. Type idl into your linux command 
line. In order to write programs choose an editor such as kwrite,emacs etc. and start the 
program with 'pro name' and finish with 'end'. To compile the program type '.compile 
name' into the idl command line, to run type in the procedure name. To exit an IDL session 
type exit. 

COMMAND LINE AND IDLDE 

The IDLdevelopment (idlde) environment provides you 5 windows: 

• Project Window: shows current project modules (if any) 
• Editing Window: where you can develop your applications 
• Output Log: shows non graphical results 
• Variable Watch Window: shows the currently existing variables their types and 

dimensionality 
• Command Line:  where you enter IDL commands 

To use the IDL interface type idlde into your linux command line, or start idlde in Windows 
by choosing it from the programs menu. You will find a command line at the bottom of the 
interface. In order to write programs use the IDL Editor. 



To use the IDL Editor, Click inside the Editor Window. You can then type in IDL 
commands, but they will not be immediately executed. When you have finished writing your 
procedure, select the 'File' option from the Menu Bar, and then select 'Save'. The name 
you choose for the file must be the same as the name of your main procedure. Then you 
can select the 'Run' option from the Menu Bar and select 'Compile (filename)'. Then if no 
errors occurred during compilation you may again select the 'Run' option from the Menu 
Bar and select 'Run (filename)'. This will execute your procedure. All output will be 
displayed in the Output Log, and any necessary input must be entered on the Input Line. 

 

• Use the stop command to stop the program at this point. You can check the 
variables in the program and if you wish to continue type .c. 

GOOD TO KNOW 

• Type ? and then the command you want to learn about for help. 
 

• IDL leaves you in the procedure when an error is detected (unless the ONERROR 
routine is called within the procedure).Enter retall to return to the top-level. 

• Array indices are zero based, e.g., 

       a = BYTARR(100) 

             FOR i=0,99 DO a(i) = i 

• If you don't use LONG(0), the maximum value is 64K, so: FOR i=LONG(0),100000 
DO sum = sum+i would give an error without the use of LONG(0). 

• > and < are not what they may seem. 

             a = 5 < 3               ; sets a to 3 (the lesser value) 

             a = WHERE (array < 2)   ; sets a = array (if first element is < 2) 

             a = WHERE (array LT 2)  ; is probably what you want 

 

 

SPECIAL CHARACTERS 

!   First character of system variable names  
 '  Delimit string constants· Indicate part of octal or hex constant  
 ;  Begin comment field  
 $ Continue on the next line· Issue operating system command if entered on a line 
by itself.  
  "  Delimit string constants or precede octal constants  



  .   Indicate constant is floating point· Start executive command  
 & Separate multiple statements on one line 
 

The most simple way to save and restore variables, system variables and IDL routines  is 
by using  SAVE and RESTORE procedures. Variables and routines cannot be saved in the 
same file.  If a set of variables are saved, they can be fully restored after reseting a 
session. The only dissadvantage is that  variables and routines are only saved in IDL 
format. The syntaxis is as follows: 

 SAVE/RESTORE 

SAVE, var1,var2,...,varn,filename='nameofthefile'  
RESTORE,nameofthefile 

 

FLOATING POINT  

IDL> x = 2. 
IDL> y = 3. 
IDL> print,' x = ',x,' y = ',y,' x/y=', x/y 
 

 idl does not require explicit variable declaration (unlike fortran or c). x=2. will assume 
that x is a floating point variable (single precision), since 2. Is (note the dot) 

IDL> x = 2 
IDL> y = 3 
IDL> print,' x = ',x,' y = ',y,' x/y=', x/y 
 
 

Unexpected? Since 2 and 3 are integers, so are x and y, hence 2/3 -> 0. Needles to say, 
this is the root of many an error!Try using the index to find which variable types exist, for 
example 

IDL>?data types   

you can also use strings 

IDL> string1 = 'Hello ' 
IDL> string2 = 'World!' 
IDL> print,string1+string2 
 



 

Most standard arithmetic functions exist, for example 

FUNCTIONS 

IDL> print,' sqrt(-1) = ',sqrt(complex(-1,0)) 
IDL> print,' atan(-1)/Pi = ',atan(-1)/!PI 

   short integers: another source of errors.try this 

IDL> i = 1 
IDL> print,' Integers: i = ',i,' i+1 = ',i+1 
 

 Good! But now try this: 

 
IDL> i = 32767 
IDL> print,' Integers: i = ',i,' i+1 = ',i+1 
 

 BAD! By default, integers in idl are short, i.e. they have values ranging between -
32768:32767. The following gives you what you intended (note the ell in 32767l) 

 
IDL> i =32767l 
IDL> print,' Long integers: i = ',i,' i+1 = ',i+1 
 

 

 
WORKING WITH ARRAYS 

Arrays may have up to 8 dimensions. The first element of an array has subscript 0, 
therefore the latest element of a given dimension should be dim-1. Arrays may be 
initialised either with all its elements equal to 0 or equal to the subscript values:  

IDL>a=indgen(5)  
IDL>b=intarr(5)  
IDL>print,'a:',a & print,'b:',b 

To extract elements, subarrays or vectors from within arrays you only need to specify the 
subscript of the element [s], the subscript boundaries [s1:s2] or specify the whole set of 
elements of a desired dimension [*] respectively. 



IDL>a=RANDOMU(seed,4,4) ; RANDOMU returns one or more uniformly-distributed, 
;floating-point, pseudo-random numbers in the range 0 < Y <1.0 to fill the array.  
IDL>print,a & print,a[0,0]  
IDL>b=a[0:2,3] & print,b  
IDL>print,a[3,*]  
IDL>print,a[3,1:*] 

Some interesting functions to play around with arrays: 

ARRAY_EQUAL (compare data for equality in situations where the index of the elements 
that differ are not of interest)  
CONGRID (resample image to any dimensions)  
INVERT (compute inverse of a square array)  
MIN  (return the minimum element of an array)  
MAX (return the maximum element of an array)  
N_ELEMENTS (number of elements of an array)  
REFORM (changes the dimensions of an array without changing the total number of 
elements)  
REPLICATE (form array of given dimensions filled with a value)  
REVERSE (reverse vectors arrays)  
ROT (rotate array by any amount)  
SHIFT (shift array elements)  
SORT (sort array contents and return vector of indices)  
REBIN (resample array by integer multiples)  
SIZE (return array size and type information, it can be seen from the Variable Watch 
Window)  
TOTAL (sum array elements)  
TRANSPOSE (Transpose array)  
WHERE (let you know for example where are the array elements satisfying a given 
condition) 

EXAMPLES: 

IDL>a = INDGEN(10)  & PRINT, 'array = ', a  
IDL>r=ARRAY_EQUAL(a,2) & print,r  
will return 0 (i.e., false) if not all the elements of a are equal to 2, otherwise will return 1, 
i.e. true) 

IDL>a=RANDOMU(seed,4,4)  & print,a  
IDL>total=N_ELEMENTS(a) & print,total 



IDL>a=RANDOMU(seed,4,4)  
IDL>print,MAX(a),MIN(a),MAX(a)/MIN(a) 

IDL>a=indgen(4) & print,a  
IDL>b = REFORM(a,2,2) & print,b 

IDL>a = [[0,1,2],[3,4,5],[6,7,8]] & print,a  
IDL>b=REVERSE(a) & print,b 

IDL>a = [[0,1,2],[3,4,5],[6,7,8]] & print,a  
IDL>b=TRANSPOSE(a) & print,b 

IDL>a = INDGEN(10)  & PRINT, 'array = ', a  
;Find the subscripts of all the elements in the array that have a value greater than 5:  
IDL> result = WHERE(a GT 6, count)  
IDL>PRINT, 'Number of elements > 6: ', count  
IDL>PRINT, 'Subscripts of elements > 6: ', result  

 

By default your plots are displayed on the IDL graphics windows. If you wish to create a 
window type: 

WINDOWS 

WINDOW [, Window_Index] [, COLORS=value] [, /FREE] [, /PIXMAP] [, RETAIN={0 | 1 | 
2}] [, TITLE=string] [, XPOS=value] [, YPOS=value] [, XSIZE=pixels] [, YSIZE=pixels] 

Of course, you do not need to specify all these keywords, but you have a number of 
possibilities described in the examples. 

EXAMPLES: 

;to create a window type  
IDL> window  
;to create a second window type  
IDL> window,1  
;if you type once again window,1 the current window will be destroyed and created once 
again. 

;To check which window is the one currently active  
IDL> whichwindow=!D.Window & print,whichwindow 



;If you wish to plot on a certain window, for example, window 5:  
IDL> WSET,5 

;To delete window 5:  
IDL> WDELETE,5  
;If you wish to delete all the window you have currently opened there is a trick:  
IDL> WHILE !D.Window ne -1 do wdelete, !D.Window 

 ;since !D.Window=-1 if no graphics window are created 

;You can change the size of your window like this:  
IDL> window,1,XSIZE=256,YSIZE=256 

;To show up your hidden window  
IDL> WSHOW,5 ;if you don't specify a number then the current window will be shown up 

;To erase the contents of the current window:  
IDL> ERASE 

;If you wish to change the color of your window  
IDL> DEVICE,DECOMPOSED=0  
IDL> ERASE,COLOR=250 ;this will make your window to look white if you had not changed 
your default color palette  
  

Plotting vectors is very simple in IDL, which allows you a large number of possibilities. The 
most important command is: 

PLOTTING 

PLOT, [X,] Y [, MAX_VALUE=value] [, MIN_VALUE=value] [, NSUM=value] [, /POLAR] [, 
THICK=value] [, /XLOG] [, /YLOG] [, /YNOZERO] 

The following Graphics Keywords are accepted by this procedure: BACKGROUND, 
CHARSIZE, CHARTHICK, CLIP, COLOR, DATA, DEVICE, FONT, LINESTYLE, NOCLIP, 
NODATA, NOERASE, NORMAL, POSITION, PSYM, SUBTITLE, SYMSIZE, T3D, THICK, 
TICKLEN, TITLE, [XYZ]CHARSIZE, [XYZ]GRIDSTYLE, [XYZ]MARGIN, [XYZ]MINOR, 
[XYZ]RANGE, [XYZ]STYLE, [XYZ]THICK, [XYZ]TICKFORMAT, [XYZ]TICKINTERVAL, 
[XYZ]TICKLAYOUT, [XYZ]TICKLEN, [XYZ]TICKNAME, [XYZ]TICKS, [XYZ]TICKUNITS, 
[XYZ]TICKV, [XYZ]TICK_GET, [XYZ]TITLE, ZVALUE. 

EXAMPLE: 



;Suppose you wish to plot a sine function:  
IDL> X=FINDGEN(21)*2*!PI/20  
IDL> PLOT,X,SIN(X) 

For adjusting the axes you should use the XSTYLE keyword and set it to 1. Other possible 
values for this keyword are 2 to extend the axis range, 4 to supress the entire axis, 8 to 
suppress the box style. 

IDL> PLOT,X,SIN(X),XSTYLE=1 

You can change line style by using the keyword LINESTYLE. The following values are 
allowed: ;0=Solid, 1=Dotted, 2=Dashed; 3=Dash Dot,  4=Dash Dot Dot , 5=Long Dashes 

IDL> PLOT,X,SIN(X),XSTYLE=1,LINESTYLE=3 ;to plot a dot dashed line 

To add the axis labels you should use XTITLE and YTITLE as well as TITLE for the title 
of your plot. 

IDL> PLOT,X,SIN(X),XSTYLE=1,LINESTYLE=3,TITLE='sin(x) plot', 
XTITLE='angle',YTITLE='sin(x)' 

To plot symbols instead of lines you should use PSYM. A number of symbols are available 
when PSYM is set from  1 to 7.  For example 1 means +, 2 means *, 7 is x, etc. When PSYM 
is set to 10 data is plotted in histogram mode. 

IDL> PLOT,X,SIN(X),XSTYLE=1,PSYM=6,TITLE='sin(x) plot', 
XTITLE='angle',YTITLE='sin(x)' ;squares  
IDL> PLOT,X,SIN(X),XSTYLE=1,PSYM=10 ;histogram 

If you wish to connect symbols with line you should set PSYM to a negative value: 

IDL> PLOT,X,SIN(X),XSTYLE=1,PSYM=-5 ;triangles with line 

The symbol size can be changed with SYMSIZE 

IDL> PLOT,X,SIN(X),XSTYLE=1,PSYM=-4,SYMSIZE=3.0 ;diamond shaped symbols will be 
3 times larger 

You can create your own symbol with USERSYM procedure. To plot your symbol you need 
to set PSYM to plus or minus 8. Symbols can be drawn with vectors or can be filled. 
Symbols can be of any size and can have up to 50 vertices. Its syntax is: 

USERSYM, X [, Y] [, COLOR=value] [, /FILL] [, THICK=value] 



The X and/or Y parameters define the vertices of the symbol as offsets from the data 
point in units of approximately the size of a character. In the case of a vector drawn 
symbol, the symbol is formed by connecting the vertices in order. 

IDL> a = [0.0, 0.5,-0.8,0.8,-0.5,0.0]  
IDL> b = [1.0,-0.8,0.3,0.3,-0.8,1.0]  
IDL> USERSYM, a,b  
IDL> PLOT,X,SIN(X),XSTYLE=1,PSYM=8,SYMSIZE=3.0 

To change the range of line plots you should use XRANGE and/or YRANGE. 

IDL> PLOT,X,SIN(X),XSTYLE=1,XRANGE=[0,2] ;Only the x values in this range will be 
plotted 

If you wish to reverse an axis, you only need to change the order of your limits in XRANGE 
and/or YRANGE. 

IDL> PLOT,X,SIN(X),XSTYLE=1,XRANGE=[4,0] 

To play around with tick marks you have several routines such as TICKLEN, 
XTICKLEN,TICKS,XMINOR. 

To enlarge the size of your labels you should use CHARSIZE. 

IDL> plot,x,sin(x),xstyle=1,linestyle=0,xtitle='x',ytitle='sin(x)',CHARSIZE=2.5 

Check the IDL help for exploiting the full capabilities offered by all these keywords.  
 

For overplotting data on your plot you can use OPLOT. This command does not generate a 
new axis, which is established by the most recent PLOT call. The syntaxis is as follows:  

OVERPLOTTING/SYMBOLS/LINES 

OPLOT, [X,] Y [, MAX_VALUE=value] [, MIN_VALUE=value] [, NSUM=value] [, /POLAR] [, 
THICK=value] 

Graphics Keywords accepted by this procedure are: [, CLIP=[X0, Y0, X1, Y1]] [, 
COLOR=value] [, LINESTYLE={0 | 1 | 2 | 3 | 4 | 5}] [, /NOCLIP] [, PSYM=integer{0 to 10}] 
[, SYMSIZE=value] [, /T3D] [, ZVALUE=value{0 to 1}] 

 



EXAMPLE: 

IDL> x=findgen(21)*2*!PI/20  
IDL> y=findgen(21)*4*!PI/20  
IDL> plot,x,sin(x),xstyle=1,linestyle=0  
IDL> OPLOT,y,sin(y),linestyle=2 

You can use the PLOTS procedure for drawing lines and symbols on your current plot. The 
syntaxis is: 

PLOTS, X [, Y [, Z]] [, /CONTINUE] 

The CONTINUE keyword is set to continue drawing a line from the last point of the most 
recent call to PLOTS. The following optional graphics keywords are available: [, CLIP=[X0, 
Y0, X1, Y1]] [, COLOR=value] [, /DATA | , /DEVICE | , /NORMAL] [, LINESTYLE={0 | 1 | 2 
| 3 | 4 | 5}] [, /NOCLIP] [, PSYM=integer{0 to 10}] [, SYMSIZE=value] [, /T3D] [, 
THICK=value] [, Z=value] 

EXAMPLE: 

IDL> x=findgen(21)*2*!PI/20  
IDL> plot,x,sin(x),xstyle=1,linestyle=0  
IDL> xvalues=[0,6]  
IDL> yvalues=[0,0]  
IDL> PLOTS,xvalues,yvalues,linestyle=1  
IDL> xvalue2=1  
IDL> yvalue2=-0.5  
IDL> PLOTS,xvalue2,yvalue2,psym=4,symsize=2.5  
  

You might write a text on your plot by using XYOUTS procedure. The syntaxis is as 
follows:  

ANNOTATION 

XYOUTS, [X, Y,] String [, ALIGNMENT=value{0.0 to 1.0}] [, CHARSIZE=value] [, 
CHARTHICK=value] [, TEXT_AXES={0 | 1 | 2 | 3 | 4 | 5}] [, width=variable] 

Particularly useful here are the keywords ORIENTATION for writing the text following a 
given angle, and ALIGNMENT, which specifies the alignment of the text baseline. An 
alignment of 0.0 (the default) aligns the left edge of the text baseline with the given (x, 



y) coordinate. An alignment of 1.0 right-justifies the text, while 0.5 results in text 
centered over the point (x, y). 

EXAMPLE: 

IDL> x=findgen(21)*2*!PI/20  
IDL> plot,x,sin(x),xstyle=1,linestyle=0  
IDL> XYOUTS,1,-0.5,'This is a sin function',orientation=45  
IDL> XYOUTS,1,-0.5,'This is a sin function',orientation=45,alignment=1 

The ANNOTATE procedure starts an IDL widget program that allows you to interactively 
annotate images and plots with text and drawings. The syntaxis is: 

ANNOTATE [, COLOR_INDICES=array] [, DRAWABLE=widget_id | , WINDOW=index] [, 
LOAD_FILE=filename] [/TEK_COLORS] 

EXAMPLE: 

IDL> ; Output an image in the current window:  
IDL> plot,x,sin(x),xstyle=1,linestyle=0  
IDL> ; Annotate it:  
IDL> ANNOTATE ;after deciding your options click on the window and save your changes 
and also your file if desired. To exit make use of the "File" option in the main menu. 

IDL provides a  large dataset of fonts. The !P.FONT variable tells IDL the font type to be 
written. When you are plotting a string you switch to the required font type by typing !. 
For example !7 specifies complex greek fonts., whereas !3 is the default simplex roman, !9 
is for Math fonts. Check the IDL help for further fonts. A useful command is 
SHOWFONT, which displays the requested vector-drawn font.. Some interesting codes 
for positioning are !a for shifting above division line, !b shift below division line !c carriage 
return, !d subscript, !l second level subscript, !e exponent, !u upper subscript. 

There is a trick to see the full set of true type font names on the system: 

IDL> device,font='*',Get_fontnames=fontnames, /tt_font  
IDL> for j=0,N_elements(fontnames)-1 do print,fontnames(j) 

EXAMPLES: 

IDL> SHOWFONT, 'Helvetica Italic', 'Helvetica Italic', /TT_FONT ;to see the whole set 
of fonts 



IDL> erase  
IDL> xyouts, 0.5,0.5, /normal,size=3, 'A!dB!n(mag)'  
IDL> erase  
IDL> xyouts, 0.5,0.5, /normal,size=3, 'Y=3x!U2!N+5x+3' 

 

!P.MULTI system variable allows you to perform multiples plots on a single page. It is a 3 
element vector [i,n,m]:  

MULTIPLE PLOTS 

i is the number of remaining plots before erasing page (i=0 when beginning a multiple plot) 

n number of horizontal sub screens 

m number of vertical sub screens 

You return to a single plot mode by typing 

!P.MULTI=0 

EXAMPLE: 

IDL> !p.multi=[0,2,2]  
IDL> x=findgen(21)*2*!PI/20  
IDL> plot,x,sin(x)  
IDL> plot,x,cos(x)  
IDL> plot,x,sin(x)/cos(x)  
IDL> plot,x,cos(x)*sin(x)  
IDL> !p.multi=0 

A full control on the location of the plots is allowed by making use of the POSITION 
keyword within the PLOT command. POSITION is a 4-element vector giving, in order, the 
coordinates [(X0, Y0), (X1, Y1)] (in normalized units ranging from 0.0 to 1.0) of the lower 
left and upper right corners of the window.  You must allow space for the annotation, 
which resides outside the window, otherwise an error message will warn you. 

Sometimes it would be interesting to set to a very small value the XCHARSIZE or 
YCHARSIZE keywords of the PLOT command to allow a single axis to be shared by two 
plots. 

 



EXAMPLE: 

IDL> ;please write and execute the following procedure called 'position_plot.pro' 

>pro position_plot  
>x=findgen(21)*2.*!pi/15.  
>y=findgen(21)*4.*!pi/15.  
>plot,x,cos(x),POSITION=[0.1,0.65,0.9,0.9],ytitle='COSENO',xtitle='X',charsize=1.2,/NO
ERASE  
>plot,x,sin(x),POSITION=[0.1,0.1,0.5,0.55],ytitle='SENO',xtitle='X',charsize=1.2,/NOERA
SE  
>plot,y,sin(y),POSITION=[0.5,0.1,0.9,0.55],YCHARSIZE=0.001,xtitle='Y',charsize=1.2,/NO
ERASE  
>end 

 

 

 

Adding colors to your plots is very simple in IDL. IDL works in two different modes with 
colors:  
   

COLORS 

• Indexed: the system has a palette of 256 RGB colors. When plotting you should 
specify the color as a number within the palette. This mode is the one appropiate 
for 8-bit graphic cards machines. This mode is activated by typing: 
DEVICE,DECOMPOSED=0. 

• RGB: the color specified as number of 24 bits, where the smallest ones refer to the 
red, the following 8 to the green and the remaining 8 to the blue. In this mode 
several millions of colors are possible. This mode is the one appropiate for 24-bit 
graphic cards machines.This mode is activated by typing: DEVICE,DECOMPOSED=1 

If you do not know whether your graphic card is 8 or 24 bits you can type the following 
commands:  

IDL> device,get_visual_name=n,get_visual_depth=d 

IDL> print,n,d 



If you have a 8-bit graphic card you only can use the indexed model, and the option 
DEVICE, DECOMPOSED=1 is not allowed. Alternatively, if you have a 24-bit graphic card 
you could work as if you were using a 8-bit machine by setting DEVICE,DECOMPOSED=0. 

You can define and load your colors with TVLCT by means of three color vectors that 
describe red, green and blue components (although other options are allowed as well). This 
is a very good approach when you only need a few colors. 

EXAMPLES: 

IDL> tvlct,[255,0,255,0,0],[255,255,0,0,0],[255,0,0,255,0],1  
 ;you have created 5 colors starting at number 1 (white: 255,255,255) and ending at 5 
(black: 0,0,0)  
IDL> device,decomposed=0  
IDL> !p.multi=[0,2,2]  
IDL> plot,hanning(21),color=1 ;hanning is a predefined function  
IDL> plot,hanning(21),color=2  
IDL> plot,hanning(21),color=3  
IDL> !p.multi=0  
IDL> plot,hanning(21),color=4,background=3 ;you are changing your background  
IDL> plot,hanning(21),color=4,background=5 

The system has a predefined set of palettes that you can load with LOADCT command. You 
need to type the number of the required palette.  XLOADCT utility provides a graphical 
widget interface to the LOADCT procedure. Moreover you can make use of the more 
clever utility XCOLORS, which belongs to the library "coyote" of D. Fanning (which has 
been installed in our local libraries directory). 

EXAMPLES: 

IDL> device,decomposed=0  
IDL> xloadct ;choose one and save it  
IDL> plot,hanning(21),color=50  
IDL> plot,hanning(21),color=150  
IDL> plot,hanning(21),color=250  
IDL> loadct,0 ;returns to the black and white linear default palette. 

 

 

 

http://www.dfanning.com/�

To generate a postscript file you have to type:  

POSTSCRIPT 

SET_PLOT,'PS' 

To enable color output you might use: 

SET_PLOT,'PS',/COPY 

or type an additional statement: 

DEVICE, COLOR=1 

To increase the number of colors in your postscript file make use of BITS_PER_PIXEL 
keyword of the DEVICE command and set it to 8: 

DEVICE, COLOR=1, BITS_PER_PIXEL=8 

You can decide the size of your output with the DEVICE keywords XSIZE and YSIZE and 
the offset keywords XOFFSET, YOFFSET. Finally, you should close your postcript file 
with 

DEVICE,/CLOSE 

To return to the default output: 

SET_PLOT,'X' 

EXAMPLE: 

IDL> set_plot,'ps'  
IDL> device,filename='mypostcript.ps',xsize=17,ysize=25,yoffset=1.  
IDL> plot,hanning(21)  
IDL> device,/close  
IDL> set_plot,'x'  

 

 

 



• http://dot.astro.uu.nl/rrweb/rjr-material/idl/manuals/idl-brief-manual.txt 

USEFUL PAGES: 

• http://www.dfanning.com/ 
• http://idlastro.gsfc.nasa.gov/ 

 

•  

This tutorial was  in parts taken from: 

http://www.iac.es/sieinvens/SINFIN/CursoIDL/idlpp5.php 
• http://www.ssel.montana.edu/HowTo/idl/tutorial.htm 
• http://dot.astro.uu.nl/rrweb/rjr-material/idl/manuals/idl-brief-manual.txt 

